



## Curve Tracer

By: Michael A. Covington  
Artificial Intelligence Center  
The University of Georgia  
Athens, GA 30602-7415  
[mc@uga.edu](mailto:mc@uga.edu)


### Equipment

- Agilent E3631A Power Supply
- Computer
- RS-232 module

### Technique

A graph of current versus voltage (an I/V curve) is a basic characteristic of almost any electronic device. Together with a PC, the Agilent E3631A power supply can measure this curve for a wide variety of devices and even capture full families of curves for power transistors.

Here's an example, the linear I/V curve of a 47-ohm resistor:






And here's the I/V curve of a red LED, which "turns on" at 1.6 volts:




These curves were captured with Program 1, shown at the end of this document. The program incorporates a couple of subtle tricks. First, since small currents are often of interest, but the measurement error of the E3631A can be up to 0.5 mA, the program determines the offset error (by applying 0 volts and measuring the current), then subtracts it from the measured values. As a result, small currents are measured more accurately, although the resolution is still only 1 mA. Second, to avoid excessive current, the program terminates each trace as soon as it goes off the top of the screen.

### Thermal Effects

It should be emphasized that the voltages and currents supplied and measured by the E3631A are suitable for relatively high-power devices; this is not the best way to test a single 1N4148 or the like. However, the testing of higher-power components can be quite revealing. Consider this I/V curve of a No. 47 lamp (a 6.3-volt, 150-mA light bulb familiar to radio repairmen of the 1950s):



As expected, the curve is concave downward, but what's the strange hook at the left? Apparently, it reflects the fact that when its heat output is relatively low, the filament doesn't have time to warm up before the measurement is taken. Here's the same curve, but this time with a 5-second delay before taking each measurement:



This time the curve is much smoother but continues to be concave downward, showing that the resistance of the bulb rises with the voltage. This nonlinearity was used in Agilent Technologies' very first product, the Model 200 audio oscillator, which used an incandescent bulb to stabilize a vacuum-tube Wien bridge.




### Transistor Curve Tracing

The E3631A can capture the family of curves of a power transistor by plotting I against V in the manner just described while applying the base current from another of its voltage outputs.


The catch is that typically, the base current needs a resolution much better than 1 mA; it's typical to step from 0 to 1 mA in five steps. Thus the base current cannot be taken directly from an output of the E3631A in current-limiting mode.

A more conventional engineer might have gotten around this by building a voltage-controlled current source (VCCS), but I wanted to keep the circuit as simple as possible, so I used the simplest VCCS possible, a resistor. Here's the complete transistor-curve-tracing circuit:




To get 1 mA to flow into the base, the computer applies a voltage of  $(4700 \text{ ohms} \times 1 \text{ mA}) + 0.6$ , where 0.6 V is the "turn-on" voltage for a silicon junction. But when trying out the circuit, I quickly discovered that the "turn-on" voltage varies from transistor to transistor. Because of this, instead of using a fixed value of 0.6 V, the program actually measures the turn-on voltage at which current first begins to flow. By doing this, it easily achieves a resolution of 0.02 mA of base current.

The program for transistor curve tracing is Program 2 at the end of this document. Here's its family of curves for a TIP33 power transistor:



And here are the curves for the same transistor, measured on a Tektronix 575 transistor curve tracer (vintage 1957) and photographed with a digital camera:



Although no substitute for the Tek 575, the E3631A, with computer control, is in some ways more versatile; besides testing transistors, it's easy to set up simple curve-tracing test jigs for FETs, thyristors, relays (measuring pull-in and drop-out voltage), and a wide variety of other components. The biggest drawback of the E3631A is that it takes a minute or more to measure each trace; with power devices, where heating is a factor, the time delay can actually make the trace more informative.

**Program 1 (HPCURVE.BAS)**

```
CLS
PRINT "Current/voltage curve of any component - M. Covington 1999"
PRINT
PRINT "Connect component under test to +25V output of Agilent E3631A."
PRINT

PRINT "Establishing communications with power supply..."
OPEN "COM2:9600,n,8,2,rs,cd,lf,pe" FOR RANDOM AS #1 LEN = 1000

PRINT #1, "SYST:REM"      ' Place E3631A in remote mode
PRINT #1, "*RST;*CLS"    ' Reset it and clear errors
PRINT #1, "*IDN?"        ' Request its ID string
LINE INPUT #1, r$
PRINT r$

PRINT
PRINT "Title of chart?"
LINE INPUT title$
INPUT "V(MAX)?", vmax
INPUT "I(MAX)?", imax

steps = 100                ' determines smoothness of graph

' Go into graphics mode

SCREEN 12
xleft = 10: xright = 630: ytop = 10: ybottom = 470
LINE (xleft, ytop)-(xright, ybottom), 4, B  ' Draw the box

PRINT
PRINT " "; title$
PRINT " V = 0 TO "; vmax
PRINT " I = 0 TO "; imax

' For greater accuracy, measure the
' zeroing error in the E3631A's current reading

PRINT #1, "APPL P25V,0,0.01"
PRINT #1, "OUTP ON"
PRINT #1, "MEAS:CURR? P25V"
LINE INPUT #1, i$
ioffset = VAL(i$)

' Ramp up the voltage, measuring the current along the way

FOR vcc = 0 TO vmax STEP vmax / steps

PRINT #1, "APPL P25V,"; vcc; ", "; imax + .001

' Time delay for stabilization --
' Needed only when heating of DUT is a factor
'SLEEP 5

PRINT #1, "MEAS:VOLT? P25V"
LINE INPUT #1, v$
v = VAL(v$)
```



```
PRINT #1, "MEAS:CURR? P25V"  
LINE INPUT #1, i$  
i = VAL(i$) - ioffset  
  
x = xleft + (v / vmax) * (xright - xleft)  
y = ybottom + (i / imax) * (ytop - ybottom)  
IF vcc = 0 THEN PSET (x, y) ELSE LINE -(x, y), 7  
  
IF i > imax THEN GOTO EndOfThisCurve ' avoid overcurrent  
  
NEXT vcc  
  
EndOfThisCurve:  
  
PRINT #1, "OUTP OFF"  
CLOSE #1  
  
' Wait for user to press Enter  
' (Cursor becomes visible when this happens)  
PRINT " ";  
LINE INPUT junk$  
END
```

**Program 2 (HPTRANS.BAS)**

```
CLS  
PRINT "TRANSISTOR CURVE TRACER - M. Covington 1999"  
PRINT  
PRINT "Tests NPN transistors using Agilent E3631A."  
PRINT "Base to +6V through 4.7K resistor."  
PRINT "Collector to +25V."  
PRINT "Emitter to ground of both supplies."  
PRINT  
  
PRINT "Establishing communications with power supply..."  
OPEN "COM2:9600,n,8,2,rs,cd,lf,pe" FOR RANDOM AS #1 LEN = 1000  
  
PRINT #1, "SYST:REM"  
PRINT #1, "**RST;*CLS"  
PRINT #1, "**IDN?"  
LINE INPUT #1, r$  
PRINT r$  
  
PRINT  
PRINT "Title of chart?"  
LINE INPUT title$  
INPUT "V(CC)(MAX) (volts) ? ", vccmax  
INPUT "I(C)(MAX) (amps) ? ", icmax  
INPUT "I(B)(MAX) (amps) ? ", ibmax  
  
steps = 100  
curves = 5  
resistor = 4700  
  
' Go into graphics mode  
  
SCREEN 12  
xleft = 10: xright = 630: ytop = 10: ybottom = 470
```



```
LINE (xleft, ytop)-(xright, ybottom), 4, B ' Draw the box

PRINT
PRINT " "; title$
PRINT " V(C) = 0 TO "; vccmax
PRINT " I(C) = 0 TO "; icmax
PRINT " I(B) = 0 TO "; ibmax

' For greater accuracy, measure the
' zeroing error in the E3631A's current reading

PRINT #1, "APPL P25V,0,0.01"
PRINT #1, "OUTP ON"
PRINT #1, "MEAS:CURR? P25V"
LINE INPUT #1, i$
ioffset = VAL(i$)

' Now determine the lowest base voltage that will cause
' detectable current to flow...

PRINT " Measuring V(BE)... ";
i = 0
v = .1
PRINT #1, "APPL P25V,"; vccmax / 2; ",0.01"
WHILE i < ioffset + .001
  v = v + .02
  PRINT #1, "APPL P6V,"; v; ",,"; ibmax + .005
  PRINT #1, "MEAS:CURR? P25V"
  LINE INPUT #1, i$
  i = VAL(i$)
WEND
vbe = v - .02 ' go back one step
PRINT vbe

' Family of curves...
FOR ib = 0 TO ibmax STEP ibmax / curves
  vb = ib * resistor + vbe
  PRINT #1, "APPL P6V,"; vb; ",,"; ibmax + .005
  IF vb > 6.1 THEN PRINT "Caution: You need a smaller resistor."

  FOR vcc = 0 TO vccmax STEP vccmax / steps
    PRINT #1, "APPL P25V,"; vcc; ",,"; icmax + .001

    ' Time delay for stabilization --
    ' Needed only when heating of DUT is a factor
    'SLEEP 5

    PRINT #1, "MEAS:VOLT? P25V"
    LINE INPUT #1, v$
    v = VAL(v$)

    PRINT #1, "MEAS:CURR? P25V"
    LINE INPUT #1, i$
    i = VAL(i$) - ioffset

    x = xleft + (v / vccmax) * (xright - xleft)
    y = ybottom + (i / icmax) * (ytop - ybottom)
    IF vcc = 0 THEN PSET (x, y) ELSE LINE -(x, y), 7

    IF i > icmax THEN GOTO EndOfThisCurve ' avoid overcurrent
```





```
NEXT vcc
EndOfThisCurve:
NEXT ib
PRINT #1, "OUTP OFF"
CLOSE #1

' Wait for user to press Enter
' (Cursor becomes visible when this happens)
PRINT " ";
LINE INPUT junk$
END
```